
==========NMOS_bsim3_SubSanity==================

NMOS_bsim3_SubSanity
Vds  3   0   dc      5v
Vgs  2   0   dc      1.2v
v1   3   1   dc      0v
m1   1   2   0   0   N1   W=3u     L=1u AD=7p AS=7p PD=10u PS=10u

.OPTIONS  GMIN=1e-15 METHOD=gear ABSTOL=1e-15 

.dc Vds 1m 5 0.01 vgs 0 1 .1

.control
run
plot  mag(i(v1)) loglog
.endc

***************SILICON_DATA****************************
*         see reference curve below

*Lmin= .35      Lmax= 20         Wmin= .6        Wmax= 20
.model          N1               NMOS
+ Level=  8        Tnom=27.0
*------------------Process----------------------------------------------- 
+ tox=160e-10      xj=0.25e-06        nch=0.5e+17                        
*------------------V_threshold------------------------------------------- 
+ vth0=0.72        nlx=0.12e-06                                          
*------------------Bulk-------------------------------------------------- 
+ k1=1.04          k2=-1.209E-01
+ cdsc=-2.4E-4     cdscd=-1.506E-04   cdscb=-2.219E-04   
*------------------mobility----------------------------------------------
+ u0=678           ua=8.964e-10                                          
+ ub=1.472e-18     uc=-4.441E-17      vsat=86000                         
*------------------Subthresshold-----------------------------------------
+ nfactor=1.8                                                               
+ cit=-5.0E-04     voff=-7.862E-02                                        
+ eta0=4.441e-16   etab=-2.E-01       dsub=0.7        
*------------------Hot electrons----------------------------------------- 
+ alpha0=1.61e-05  beta0=36.68                                           
*------------------VAF---------------------------------------------------
+ lint=.12e-06     pclm=.19           pscbe1=3.79e+08   pscbe2=9.4e-05   
+ delta=0.01655    pvag=0.4484    
*------------------Bulk_diode-------------------------------------------- 
+ js=5.858e-08    
*------------------Resistance-------------------------------------------- 
+ rsh=70           rdsw=375                                                
+ wr=0.7586        prwb=0             prwg=-4.441E-17    
*------------------Capacitance-------------------------------------------
+ cj=0.0002424     cjsw=2.73e-10      mj=0.3551           mjsw=0.3873
+ cgso=9e-13       cgdo=9e-13         cgbo=7e-10  
+ pb=0.5614        pbsw=0.8           xpart=0                             



+ dlc=5e-08        dwc=1.5e-07  
*------------------BulkChargeEffect--------------------------------------  
+ a0=0.7           a1=0                a2=1                ags=0.05583    
+ b0=6.305e-08     b1=6.579e-08        keta=-1.531E-02                    
*------------------ShortChannel------------------------------------------ 
+ dvt0=2.2         dvt1=0.53          dvt2=-1.521E-01      drout=0.76       
+ pdiblcb=.4       pdiblc1=0.00886    pdiblc2=0.00029                      
*------------------NarrowChannel-----------------------------------------
+ w0=2.6e-04       wint=0.16e-06                                          
+ ww=-9.525E-14    wwn=1.0                                               
+ dvt0w=0          dvt1w=5.3e6        dvt2w=-1.E-01                        
+ k3=2.53          k3b=-5             dwg=0                dwb=0  
*------------------Noise-------------------------------------------------
+ af=1             kf=1e-28           ef=0.95                            
*------------------Temperature------------------------------------------- 
+ pvsat=0          ute=-1.258E+00     kt1=-3.85E-01                      
+ kt1l=0           kt2=-3.098E-02     ua1=5.705e-09                      
+ ub1=-1.147E-17   uc1=-1.302E-01     at=20380                           
* prt=-3.287E+02   lk1=0              lk2=0                              
+ lvsat=0          la0=0              lags=0             lute=0          
+ luc=0            

*Lmin= .35      Lmax= 20         Wmin= .6        Wmax= 20
.model          N2               NMOS
+Level=  8      Tnom=27.0
****************Process***********************
+Tox=16e-9      Xj=250E-09       Nch=.5E+17
****************V_thresshold***********************
+Vth0=.72       Nlx=0.12E-06
****************Bulk***********************
+K1=1.04        K2=-1.2e-1       K3=-2.612 
****************Mobility***********************
+U0=478         Ua=6.47e-9       
+Ub=4.23e-18    Uc=-4.706281E-11 Vsat=86301
****************Subthresshold***********************
+NFactor=1.14   
+Cit= 1.6E-04   Voff=-6.74E-02  
+Eta0=1.03E-02  Etab=-5.04E-03   Dsub= .32
****************VAF***********************
+Lint=.12e-6    Pclm=.19         Pscbe1=4e8      Pscbe2=5E-09
+delta=0.01     Pvag=.44
****************Resistance***********************
+Rdsw=650       Prwg =0          Prwb = -.213    wr=1
****************Capacitance***********************
+Cdsc=-2.15E-05 Cdscb=0          Cdscd=0
****************BulkChargeEfffect***********************
+A0=.35         A1= 2.8E-02      A2= .9          Ags=.1    
+B0=0.546       B1=1             Keta=-3.6E-02
****************ShortChannel*********************** 
+Dvt0=2.812     Dvt1= 0.462      Dvt2=-9.2e-2    Drout= .31871233 
+Pdiblcb= -.234 Pdiblc1= 2.5E-03 Pdiblc2=6.4E-03 
****************NarrowChannel***********************
+W0=1.163e-6    Wint=1.47e-7     Ww=-1.42E-09    Wwn=.2613948
+Wl=0           Wwl=0            Wln=0 
+Dvt0w=0        Dvt1w=0          Dvt2w=0 
+K3b= 2.233     Dwg=-6.0E-09     Dwb= -3.56E-09
****************Process***********************
+Ll=1.30E-10    Lw=0             Lwl=0 
+Lln=.316394    Lwn=0
+Ute=-1.48 
****************Temperature***********************
+Kt1l=0         kt1=-.3          kt2=-.051     
+Ua1= 3.31E-10  Ub1= 2.61E-19    Uc1= -3.42e-10 
+At= 22400      Prt=764.3

.model             NMOSC              NMOS
+ Level=  8        Tnom=27.0
*------------------Process----------------------------------------------- 
+ tox=160e-10      xj=0.25e-06        nch=0.5e+17                        
*------------------V_threshold------------------------------------------- 



+ vth0=0.72        nlx=0.12e-06                                          
*------------------Bulk-------------------------------------------------- 
+ k1=1.04          k2=-1.209E-01
*------------------mobility----------------------------------------------
+ u0=678           ua=8.964e-10                                          
+ ub=1.472e-18     uc=-4.441E-17      vsat=86000                         
*------------------Subthresshold-----------------------------------------
+ nfactor=1.8                                                               
+ cit=-5.0E-04     voff=-7.862E-02                                        
+ eta0=4.441e-16   etab=-2.E-01       dsub=0.7                           
*------------------Hot electrons----------------------------------------- 
* alpha0=1.61e-05  beta0=36.68                                           
*------------------VAF---------------------------------------------------
+ lint=.12e-06     pclm=.19           pscbe1=3.79e+08   pscbe2=9.4e-05   
+ delta=0.01655    pvag=0.4484                                            
*------------------Bulk_diode-------------------------------------------- 
* + js=5.858e-08                                                              
*------------------Resistance-------------------------------------------- 
+ rsh=70           rdsw=375                                                
+ wr=0.7586        prwb=0             prwg=-4.441E-17                    
*------------------Capacitance-------------------------------------------
+ cj=0.0002424     cjsw=2.73e-10      mj=0.3551           mjsw=0.3873    
+ cgso=9e-13       cgdo=9e-13         cgbo=7e-10                         
+ cdsc=-2.4E-4     cdscd=-1.506E-01   cdscb=-2.219E-04                   
+ pb=0.5614        pbsw=0.8           xpart=0                             
+ dlc=5e-08        dwc=1.5e-07  
*------------------BulkChargeEffect--------------------------------------  
* a0=0.7           a1=0                a2=1                ags=0.05583    
* b0=6.305e-08     b1=6.579e-08        keta=-1.531E-02                    
*------------------ShortChannel------------------------------------------ 
+ dvt0=2.2         dvt1=0.53          dvt2=-1.521E-01      drout=0.76       
+ pdiblcb=.4       pdiblc1=0.00886    pdiblc2=0.00029                      
*------------------NarrowChannel-----------------------------------------
+ w0=2.6e-04       wint=0.16e-06                                          
+ ww=-9.525E-14    wwn=1.0                                               
+ dvt0w=0          dvt1w=5.3e6        dvt2w=-1.E-01                        
+ k3=2.53          k3b=-5             dwg=0                dwb=0  
*------------------Noise-------------------------------------------------
* af=1             kf=1e-28           ef=0.95                            
*------------------Temperature------------------------------------------- 
* pvsat=0          ute=-1.258E+00     kt1=-3.85E-01                      
* kt1l=0           kt2=-3.098E-02     ua1=5.705e-09                      
* ub1=-1.147E-17   uc1=-1.302E-01     at=20380                           
* prt=-3.287E+02   lk1=0              lk2=0                              
* lvsat=0          la0=0              lags=0             lute=0          
* luc=0            
                                       

.end





****************SILICON_DATA*****************************
*
*         NMOS_Subthresshold_Drain_Current and Gate_Voltage
*    1mA  _______________________________________________
*        |           .           .           .           .
*        |  w=20um   . l = 1u    .           .           .
*        |           .           .           .           .
*  100uA |.......................................960mV....
*        |           .           .        960mV          .
*        |           .         960mV         .   840mV   .
*        |           .           .        840mV          .
*   10uA |.....................840mV......................
*        |          960mV        .           .           .
*        |           .           .           .   720mV   .
*        |          840mV      720mV       720mV         .
*    1uA |................................................
*        |           .           .           .           .
*        |960mV     720mV        .           .           .
*        |840mV      .           .           .   600mV   .
*  100nA |.....................600mV......600mV...........
*        |           .           .           .           .
*        |720mV     600mV        .           .           .
*        |           .           .           .   480mV   .
*   10nA |................................480mV...........
*        |           .         480mV         .           .
*        |600mV     480mV        .           .           .
*        |           .           .           .           .
*    1nA |.......................................360mV....
*        |           .           .        360mV          .
*        |           .         360mV         .           .
*        |           .           .           .           .
*  100pA |480mV.....360mV........................240mV....
*        |           .           .           .           .
*        |           .           .        240mV          .
*        |           .         240mV         .           .
*   10pA |360mV..................................120mV....
*        |           .           .           .           .
*        |          240mV        .           .           .
*        |           .           .           .           .
*    1pA |..240mV..............120mV......120mV...0mV.....
*        |           .           .           .           .
*        |          120mV        .          0mV          .
*        |           .           .           .           .
*  100fA |__120mV__________________0mV___________________.
*        1mV        10mV        100mV       1V         10V
*
*                         Drain Voltage


